PrSAT Tutorial (*Mathematica* Notebook)

June 16, 2011

First, load in the `PrSAT` package

See my **PrSAT website** for instructions on downloading and installing **PrSAT** (assuming you have *Mathematica* installed).

```
<< PrSAT`
```

Example #1

The first example of a probability model that we saw was the following:

$$\texttt{MODEL1 = PrSAT}\left[\left\{\Pr[X \wedge Y] == \frac{1}{6},\ \Pr[X \wedge \neg Y] == \frac{1}{4},\ \Pr[\neg X \wedge Y] == \frac{1}{8},\ \Pr[\neg X \wedge \neg Y] == \frac{11}{24}\right\}\right]$$

$$\left\{\{X \to \{a_2, a_4\}, Y \to \{a_3, a_4\}, \Omega \to \{a_1, a_2, a_3, a_4\}\}, \left\{a_1 \to \frac{11}{24}, a_2 \to \frac{1}{4}, a_3 \to \frac{1}{8}, a_4 \to \frac{1}{6}\right\}\right\}$$

PrSAT will show us an STT representation of **MODEL1**:

```
TruthTable[MODEL1]
```

X	Y	var	Pr
T	T	a_4	$\frac{1}{6}$
T	F	a_2	$\frac{1}{4}$
F	T	a_3	$\frac{1}{8}$
F	F	a_1	$\frac{11}{24}$

We can use **PrSAT** to calculate probability, using **MODEL1**:

$$\texttt{EvaluateProbability}[\{\Pr[X \vee Y], \Pr[X], \Pr[Y]\}, \texttt{MODEL1}]$$

$$\left\{\frac{13}{24}, \frac{5}{12}, \frac{7}{24}\right\}$$

We can also check arbitrary claims to see if they are *true on* **MODEL1**:

$$\texttt{EvaluateProbability}[\Pr[X \mid Y] > \Pr[X], \texttt{MODEL1}]$$

```
True
```

Example #2

The second example we saw was an algebraic proof of the following theorem:

$$\mathbf{Pr}(X \vee Y) = \mathbf{Pr}(X) + \mathbf{Pr}(Y) - \mathbf{Pr}(X \wedge Y)$$

```
Pr[X || Y] == Pr[X] + Pr[Y] - Pr[X && Y]
```

PrSAT easily verifies this theorem (note: it does not present a readable proof).

```
PrSAT[{Pr[X ∨ Y] ≠ Pr[X] + Pr[Y] - Pr[X ∧ Y]}]
```

PrSAT::srchfail : Search phase failed; attempting FindInstance

```
{}
```

This output means there are no probability models on which **Pr[X ∨ Y] ≠ Pr[X] + Pr[Y] - Pr[X ∧ Y]**. That "proves" that the above statement is a theorem of probability calculus.

Example #3

The second example we saw was an algebraic proof of the following theorem:

$\mathbf{Pr}(X) = \mathbf{Pr}(X \wedge Y) + \mathbf{Pr}(X \wedge \neg Y)$

```
Pr[X] == Pr[X && Y] + Pr[X && ¬ Y]
```

PrSAT easily verifies this theorem (note: it does not present a readable proof).

```
PrSAT[{Pr[X] ≠ Pr[X ∧ Y] + Pr[X ∧ ¬ Y]}]
```

PrSAT::srchfail : Search phase failed; attempting FindInstance

```
{}
```

This output means there are no probability models on which **Pr[X] ≠ Pr[X ∧ Y] + Pr[X ∧ ¬ Y]**. That "proves" that the above statement is a theorem of probability calculus.

Example #4

The next example involves the following theorem:

$\mathbf{Pr}(X \to Y) \geq \mathbf{Pr}(Y \mid X)$

```
Pr[X → Y] ≥ Pr[Y | X]
```

PrSAT easily verifies this theorem (note: it does not present a readable proof). First, we need to define the conditional operator.

```
X_ → Y_ := ¬ X ∨ Y;
```

```
PrSAT[{Pr[X → Y] < Pr[Y | X]}]
```

PrSAT::srchfail : Search phase failed; attempting FindInstance

```
{}
```

This output means there are no probability models on which **Pr[X → Y] ≥ Pr[Y | X]**. That "proves" that the above statement is a theorem of probability calculus.

Example #5

The next example involves the following theorem:

$d(X, Y) = d(X \vee Y, Y) + d(X \vee \neg Y, Y)$, **where** $d(X, Y) = \mathbf{Pr}(X \mid Y) - \mathbf{Pr}(X)$.

PrSAT easily verifies this theorem (note: it does not present a readable proof). First, we need to define **d(X, Y)**.

```
d[X_, Y_] := Pr[X | Y] - Pr[X];
```

```
PrSAT[{d[X, Y] ≠ d[X ∨ Y, Y] + d[X ∨ ¬ Y, Y]}]
```

PrSAT::srchfail : Search phase failed; attempting FindInstance

```
{}
```

This output means there are no probability models on which **d[X, Y] ≠ d[X ∨ Y, Y] + d[X ∨ ¬ Y, Y]**. That "proves" that the above statement is a theorem of probability calculus.

Example #6

The next example involves the fact that Pr(X | Y ∨ Z) = Pr(X | Y ∧ Z) is NOT a theorem:

PrSAT easily finds a counter-model to this claim.

```
PrSAT[{Pr[X | Y ∨ Z] ≠ Pr[X | Y ∧ Z]}]
```

$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to \frac{63}{125528}, a_2 \to 0, a_3 \to \frac{1}{568}, a_4 \to \frac{110}{221}, a_5 \to 0, a_6 \to 0, a_7 \to 0, a_8 \to \frac{1}{2}\}\}$

The model **PrSAT** finds by defualt is *non*-regular. We can force it to find a *regular* counter-model, as follows:

```
MODEL2 = PrSAT[{Pr[X | Y ∨ Z] ≠ Pr[X | Y ∧ Z]}, Probabilities → Regular]
```

$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to \frac{6053}{6111882}, a_2 \to \frac{1}{999}, a_3 \to \frac{1}{999}, a_4 \to \frac{1}{999}, a_5 \to \frac{1}{46}, a_6 \to \frac{10}{21}, a_7 \to \frac{85}{171}, a_8 \to \frac{1}{999}\}\}$

Here is an STT representation of **MODEL2**:

```
TruthTable[MODEL2]
```

X	Y	Z	var	Pr
T	T	T	a_8	$\frac{1}{999}$
T	T	F	a_5	$\frac{1}{46}$
T	F	T	a_6	$\frac{10}{21}$
T	F	F	a_2	$\frac{1}{999}$
F	T	T	a_7	$\frac{85}{171}$
F	T	F	a_3	$\frac{1}{999}$
F	F	T	a_4	$\frac{1}{999}$
F	F	F	a_1	$\frac{6053}{6111882}$

We can calculate the values of **Pr[X | Y ∧ Z]**, **Pr[X | Y ∨ Z]** on this model as follows:

```
EvaluateProbability[{Pr[X | Y ∧ Z], Pr[X | Y ∨ Z]}, MODEL2]
```

$\{\frac{19}{9454}, \frac{3049405}{6099711}\}$

We can look at decimal representations of these exact real numbers, as follows:

```
% // N
```

{0.00200973, 0.499926}

We gave a different model in the lecture notes. We can enter that model in by hand, and then verify it has the desired properties, as follows:

```
MODEL3 =
 PrSAT[{Pr[X ∧ Y ∧ Z] == 1/6, Pr[X ∧ Y ∧ ¬ Z] == 1/6, Pr[X ∧ ¬ Y ∧ Z] == 1/4, Pr[X ∧ ¬ Y ∧ ¬ Z] == 1/16,
   Pr[¬ X ∧ Y ∧ Z] == 1/6, Pr[¬ X ∧ Y ∧ ¬ Z] == 1/12, Pr[¬ X ∧ ¬ Y ∧ Z] == 1/24, Pr[¬ X ∧ ¬ Y ∧ ¬ Z] == 1/16}]
```

$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\},$
$Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$
$\{a_1 \to \frac{1}{16}, a_2 \to \frac{1}{16}, a_3 \to \frac{1}{12}, a_4 \to \frac{1}{24}, a_5 \to \frac{1}{6}, a_6 \to \frac{1}{4}, a_7 \to \frac{1}{6}, a_8 \to \frac{1}{6}\}\}$

```
TruthTable[MODEL3]
```

X	Y	Z	var	Pr
T	T	T	a_8	$\frac{1}{6}$
T	T	F	a_5	$\frac{1}{6}$
T	F	T	a_6	$\frac{1}{4}$
T	F	F	a_2	$\frac{1}{16}$
F	T	T	a_7	$\frac{1}{6}$
F	T	F	a_3	$\frac{1}{12}$
F	F	T	a_4	$\frac{1}{24}$
F	F	F	a_1	$\frac{1}{16}$

```
EvaluateProbability[{Pr[X | Y ∧ Z], Pr[X | Y ∨ Z]}, MODEL3]
```

$\{\frac{1}{2}, \frac{2}{3}\}$

We can also see the algebraic form of an expression, as follows:

```
AlgebraicForm[Pr[X | Y ∧ Z] == Pr[X | Y ∨ Z], {X, Y, Z}]
```

$$\frac{a_8}{a_7 + a_8} == \frac{a_5 + a_6 + a_8}{a_3 + a_4 + a_5 + a_6 + a_7 + a_8}$$

Note that **PrSAT** uses different conventions (*i.e.*, a different ordering in the truth-table) for the a_i than I use in the lecture notes.

Here is a verifcation of some of our *independence* theorems:

```
PrSAT[{
  Pr[P ⋀ Q] == Pr[P] Pr[Q],
  Pr[P | Q] ≠ Pr[P] ,
 }]
```

PrSAT::srchfail : Search phase failed; attempting FindInstance

{}

```
PrSAT[{
  Pr[P ⋀ ¬ Q] == Pr[P] Pr[¬ Q],
  Pr[P ⋀ Q] ≠ Pr[P] Pr[Q]
 }]
```

PrSAT::srchfail : Search phase failed; attempting FindInstance

{}

```
PrSAT[{
   Pr[P ⋀ Q] == Pr[P] Pr[Q],
   Pr[P ⋀ ¬ Q] ≠ Pr[P] Pr[¬ Q]
  }]
```

PrSAT::srchfail : Search phase failed; attempting FindInstance

{}

Here is a (regular) model showing that pairwise independence does *not* entail mutual independence (for a set of 3 events)

```
PrSAT[{
   Pr[X ∧ Y] == Pr[X] Pr[Y],
   Pr[X ∧ Z] == Pr[X] Pr[Z],
   Pr[Y ∧ Z] == Pr[Y] Pr[Z],
   Pr[(X ∧ Y) ∧ Z] ≠ Pr[X ∧ Y] Pr[Z],
   Pr[X] == Pr[Y] == Pr[Z] == 1 / 2
  }, Probabilities → Regular]
```

$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to \frac{1}{676}, a_2 \to \frac{42}{169}, a_3 \to \frac{42}{169}, a_4 \to \frac{42}{169}, a_5 \to \frac{1}{676}, a_6 \to \frac{1}{676}, a_7 \to \frac{1}{676}, a_8 \to \frac{42}{169}\}\}$

```
TruthTable[%]
```

X	Y	Z	var	Pr
T	T	T	a_8	$\frac{42}{169}$
T	T	F	a_5	$\frac{1}{676}$
T	F	T	a_6	$\frac{1}{676}$
T	F	F	a_2	$\frac{42}{169}$
F	T	T	a_7	$\frac{1}{676}$
F	T	F	a_3	$\frac{42}{169}$
F	F	T	a_4	$\frac{42}{169}$
F	F	F	a_1	$\frac{1}{676}$

Here is a (regular) model showing that independence is *not* transitive:

```
PrSAT[{
   Pr[X ∧ Y] == Pr[X] Pr[Y],
   Pr[Y ∧ Z] == Pr[Y] Pr[Z],
   Pr[X ∧ Z] ≠ Pr[X] Pr[Z],
   Pr[X] == Pr[Y] == Pr[Z] == 1 / 2
  }, Probabilities → Regular]
```

$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\}, Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to \frac{1}{999}, a_2 \to \frac{995}{3996}, a_3 \to \frac{1}{999}, a_4 \to \frac{995}{3996}, a_5 \to \frac{995}{3996}, a_6 \to \frac{1}{999}, a_7 \to \frac{995}{3996}, a_8 \to \frac{1}{999}\}\}$

```
TruthTable[%]
```

X	Y	Z	var	Pr
T	T	T	a_8	$\frac{1}{999}$
T	T	F	a_5	$\frac{995}{3996}$
T	F	T	a_6	$\frac{1}{999}$
T	F	F	a_2	$\frac{995}{3996}$
F	T	T	a_7	$\frac{995}{3996}$
F	T	F	a_3	$\frac{1}{999}$
F	F	T	a_4	$\frac{995}{3996}$
F	F	F	a_1	$\frac{1}{999}$

Extra Example (not from my lecture #1 notes): Here's an example of a set of constraints that are satisfiable but only on probability models containing *irrational* numbers

$$\text{PrSAT}\left[\left\{\Pr[Y \mid X] == \Pr[Y \vee X],\ \Pr[X \wedge Y] == \frac{1}{4},\ \Pr[\neg X \wedge Y] == \frac{1}{4}\right\}\right]$$

$$\left\{\{X \to \{a_2, a_4\},\ Y \to \{a_3, a_4\},\ \Omega \to \{a_1, a_2, a_3, a_4\}\},\ \left\{a_1 \to \frac{1}{8}\left(7 - \sqrt{17}\right),\ a_2 \to \frac{1}{8}\left(-3 + \sqrt{17}\right),\ a_3 \to \frac{1}{4},\ a_4 \to \frac{1}{4}\right\}\right\}$$

```
TruthTable[%]
```

X	Y	var	Pr
T	T	a_4	$\frac{1}{4}$
T	F	a_2	$\frac{1}{8}\left(-3 + \sqrt{17}\right)$
F	T	a_3	$\frac{1}{4}$
F	F	a_1	$\frac{1}{8}\left(7 - \sqrt{17}\right)$

Can you prove that this is the *only* model for this set of constraints?

The first numerical probability model appearing in the lecture notes

$$\text{MODEL1} = \text{PrSAT}\left[\left\{\Pr[X \wedge Y] == \frac{1}{6},\ \Pr[X \wedge \neg Y] == \frac{1}{4},\ \Pr[\neg X \wedge Y] == \frac{1}{8}\right\}\right]$$

$$\left\{\{X \to \{a_2, a_4\},\ Y \to \{a_3, a_4\},\ \Omega \to \{a_1, a_2, a_3, a_4\}\},\ \left\{a_1 \to \frac{11}{24},\ a_2 \to \frac{1}{4},\ a_3 \to \frac{1}{8},\ a_4 \to \frac{1}{6}\right\}\right\}$$

```
TruthTable[MODEL1]
```

X	Y	var	Pr
T	T	a_4	$\frac{1}{6}$
T	F	a_2	$\frac{1}{4}$
F	T	a_3	$\frac{1}{8}$
F	F	a_1	$\frac{11}{24}$

Using PrSAT to translate probabilistic expressions into algebraic ones

```
AlgebraicForm[Pr[¬ X ∨ Y] < Pr[Y | X], {X, Y}]
```

$$1 - a_2 < \frac{a_4}{a_2 + a_4}$$

```
DenialOfTheorem = AlgebraicForm[Pr[¬ X ∨ Y] < Pr[Y | X], {X, Y}]
```

$$1 - a_2 < \frac{a_4}{a_2 + a_4}$$

We can then use *Mathematica*'s decision procedure for real algebra (**FindInstance**) to *refute* this claim, assuming that the underlying variables are constrained so as to constitute a *probability model*:

```
FindInstance[DenialOfTheorem && a1 + a2 + a3 + a4 == 1 &&
  0 ≤ a1 ≤ 1 && 0 ≤ a2 ≤ 1 && 0 ≤ a3 ≤ 1 && 0 ≤ a4 ≤ 1, {a1, a2, a3, a4}]
```

{}

Using PrSAT to find a model on which Pr(X | Y ∧ Z) ≠ Pr(X | Y ∨ Z)

```
MODEL2 = PrSAT[{
    Pr[X | Y ∧ Z] ≠ Pr[X | Y ∨ Z]
  },
  Probabilities → Regular, BypassSearch → True]
```

$\{\{X \to \{a_2, a_5, a_6, a_8\}, Y \to \{a_3, a_5, a_7, a_8\},$
$Z \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\},$
$\{a_1 \to \frac{1}{16}, a_2 \to \frac{1}{16}, a_3 \to \frac{1}{4}, a_4 \to \frac{5}{16}, a_5 \to \frac{1}{32}, a_6 \to \frac{1}{8}, a_7 \to \frac{1}{8}, a_8 \to \frac{1}{32}\}\}$

```
TruthTable[MODEL2]
```

X	Y	Z	var	Pr
T	T	T	a_8	$\frac{1}{32}$
T	T	F	a_5	$\frac{1}{32}$
T	F	T	a_6	$\frac{1}{8}$
T	F	F	a_2	$\frac{1}{16}$
F	T	T	a_7	$\frac{1}{8}$
F	T	F	a_3	$\frac{1}{4}$
F	F	T	a_4	$\frac{5}{16}$
F	F	F	a_1	$\frac{1}{16}$

```
EvaluateProbability[{Pr[X | Y ∧ Z], Pr[X | Y ∨ Z]}, MODEL2]
```

$\{\frac{1}{5}, \frac{3}{14}\}$

Using **PrSAT** to show the ordinal *non*-equivalence of $\{d, r, \ell, s\}$

```
d[ℍ_, 𝔼_] := Pr[ℍ | 𝔼] - Pr[ℍ];
r[ℍ_, 𝔼_] := (Pr[ℍ | 𝔼] - Pr[ℍ]) / (Pr[ℍ | 𝔼] + Pr[ℍ]);
ℓ[ℍ_, 𝔼_] := (Pr[𝔼 | ℍ] - Pr[𝔼 | ¬ ℍ]) / (Pr[𝔼 | ℍ] + Pr[𝔼 | ¬ ℍ]);
s[ℍ_, 𝔼_] := Pr[ℍ | 𝔼] - Pr[ℍ | ¬ 𝔼];

J[ℍ_, 𝔼_] := 1/Pr[¬ ℍ | 𝔼] - 1/Pr[¬ ℍ];
```

A model showing that *d* and *r* are not ordinally equivalent:

```
MODEL3 = PrSAT[{
    d[ℍ1, 𝔼1] ≥ d[ℍ2, 𝔼2],
    r[ℍ1, 𝔼1] < r[ℍ2, 𝔼2]
  },
  Probabilities → Regular]
```

$\{\{\mathbb{E}_1 \to \{a_2, a_6, a_7, a_8, a_{12}, a_{13}, a_{14}, a_{16}\}, \mathbb{E}_2 \to \{a_3, a_6, a_9, a_{10}, a_{12}, a_{13}, a_{15}, a_{16}\},$
$\mathbb{H}_1 \to \{a_4, a_7, a_9, a_{11}, a_{12}, a_{14}, a_{15}, a_{16}\}, \mathbb{H}_2 \to \{a_5, a_8, a_{10}, a_{11}, a_{13}, a_{14}, a_{15}, a_{16}\},$
$\Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8, a_9, a_{10}, a_{11}, a_{12}, a_{13}, a_{14}, a_{15}, a_{16}\}\},$
$\{a_1 \to \frac{45\,122\,521\,985\,018\,694\,529}{395\,712\,164\,128\,624\,041\,600}, a_2 \to \frac{1}{979}, a_3 \to \frac{4}{27}, a_4 \to \frac{6}{19}, a_5 \to \frac{1}{999},$
$a_6 \to \frac{1}{985}, a_7 \to \frac{1}{998}, a_8 \to \frac{2}{45}, a_9 \to \frac{1}{116}, a_{10} \to \frac{28}{141}, a_{11} \to \frac{1}{384},$
$a_{12} \to \frac{11}{100}, a_{13} \to \frac{1}{447}, a_{14} \to \frac{1}{165}, a_{15} \to \frac{2}{45}, a_{16} \to \frac{1}{999}\}\}$

```
TruthTable[MODEL3]
```

$\mathbb{E}_1$	$\mathbb{E}_2$	$\mathbb{H}_1$	$\mathbb{H}_2$	var	Pr
T	T	T	T	a_{16}	$\frac{1}{999}$
T	T	T	F	a_{12}	$\frac{11}{100}$
T	T	F	T	a_{13}	$\frac{1}{447}$
T	T	F	F	a_6	$\frac{1}{985}$
T	F	T	T	a_{14}	$\frac{1}{165}$
T	F	T	F	a_7	$\frac{1}{998}$
T	F	F	T	a_8	$\frac{2}{45}$
T	F	F	F	a_2	$\frac{1}{979}$
F	T	T	T	a_{15}	$\frac{2}{45}$
F	T	T	F	a_9	$\frac{1}{116}$
F	T	F	T	a_{10}	$\frac{28}{141}$
F	T	F	F	a_3	$\frac{4}{27}$
F	F	T	T	a_{11}	$\frac{1}{384}$
F	F	T	F	a_4	$\frac{6}{19}$
F	F	F	T	a_5	$\frac{1}{999}$
F	F	F	F	a_1	$\frac{45\,122\,521\,985\,018\,694\,529}{395\,712\,164\,128\,624\,041\,600}$

```
EvaluateProbability[{{d[ℍ₁, 𝔼₁], d[ℍ₂, 𝔼₂]},
    {r[ℍ₁, 𝔼₁], r[ℍ₂, 𝔼₂]}}, MODEL3] // N
```

{{0.21837, 0.178694}, {0.182368, 0.229258}}

A model showing that *d* and *ℓ* are not ordinally equivalent:

```
MODEL4 = PrSAT[{
    d[ℍ₁, 𝔼₁] ≥ d[ℍ₂, 𝔼₂],
    ℓ[ℍ₁, 𝔼₁] < ℓ[ℍ₂, 𝔼₂]
  },
  Probabilities → Regular]
```

$$\{\{\mathbb{E}_1 \to \{a_2, a_6, a_7, a_8, a_{12}, a_{13}, a_{14}, a_{16}\}, \mathbb{E}_2 \to \{a_3, a_6, a_9, a_{10}, a_{12}, a_{13}, a_{15}, a_{16}\}, \mathbb{H}_1 \to \{a_4, a_7, a_9, a_{11}, a_{12}, a_{14}, a_{15}, a_{16}\}, \mathbb{H}_2 \to \{a_5, a_8, a_{10}, a_{11}, a_{13}, a_{14}, a_{15}, a_{16}\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8, a_9, a_{10}, a_{11}, a_{12}, a_{13}, a_{14}, a_{15}, a_{16}\}\},$$

$$\{a_1 \to \frac{1\,525\,917\,434\,359}{4\,455\,027\,404\,640}, a_2 \to \frac{1}{113}, a_3 \to \frac{1}{21}, a_4 \to \frac{1}{129}, a_5 \to \frac{1}{132}, a_6 \to \frac{1}{38}, a_7 \to \frac{1}{30}, a_8 \to \frac{1}{32}, a_9 \to \frac{1}{28}, a_{10} \to \frac{1}{40}, a_{11} \to \frac{1}{138}, a_{12} \to \frac{1}{20}, a_{13} \to \frac{1}{9}, a_{14} \to \frac{1}{47}, a_{15} \to \frac{1}{496}, a_{16} \to \frac{8}{33}\}\}$$

I leave the remaining pairs as exercises.

Selection of Properties (1)-(8)

Verification that *r satisfies commutativity* (4):

```
PrSAT[{

  r[ℍ, 𝔼] ≠ r[𝔼, ℍ]

 }, BypassSearch → True]
```

```
{}
```

Model showing that *s violates* (4):

```
MODEL5 = PrSAT[{

   Pr[ℍ | 𝔼] > Pr[ℍ],
   s[ℍ, 𝔼] ≠ s[𝔼, ℍ]

  }, Probabilities → Regular]
```

$\{\{\mathbb{E} \to \{a_2, a_4\}, \mathbb{H} \to \{a_3, a_4\}, \Omega \to \{a_1, a_2, a_3, a_4\}\}, \{a_1 \to \frac{16511}{27972}, a_2 \to \frac{1}{4}, a_3 \to \frac{1}{999}, a_4 \to \frac{10}{63}\}\}$

```
TruthTable[MODEL5]
```

𝔼	ℍ	var	Pr
T	T	a_4	$\frac{10}{63}$
T	F	a_2	$\frac{1}{4}$
F	T	a_3	$\frac{1}{999}$
F	F	a_1	$\frac{16511}{27972}$

```
EvaluateProbability[{s[ℍ, 𝔼], s[𝔼, ℍ]}, MODEL5] // N
```

```
{0.386657, 0.696209}
```

Model showing that ℓ *violates* the so-called "Law of Likelihood" (2):

```
MODEL6 = PrSAT[{

   Pr[𝔼 | ℍ1] > Pr[𝔼 | ℍ2],
   ℓ[ℍ1, 𝔼] < ℓ[ℍ2, 𝔼]

  }, Probabilities → Regular]
```

$\{\{\mathbb{E} \to \{a_2, a_5, a_6, a_8\}, \mathbb{H}1 \to \{a_3, a_5, a_7, a_8\}, \mathbb{H}2 \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to \frac{71843009866}{96210749943}, a_2 \to \frac{3}{43}, a_3 \to \frac{10}{99}, a_4 \to \frac{1}{177}, a_5 \to \frac{1}{189}, a_6 \to \frac{1}{629}, a_7 \to \frac{2}{29}, a_8 \to \frac{1}{999}\}\}$

Model showing that *s violates* (3) — people were very surprised when I reported this one:

```
MODEL7 = PrSAT[{

   Pr[ℍ | 𝔼1] > Pr[ℍ | 𝔼2],
   s[ℍ, 𝔼1] < s[ℍ, 𝔼2]

  }, Probabilities → Regular]
```

$\{\{\mathbb{E}1 \to \{a_2, a_5, a_6, a_8\}, \mathbb{E}2 \to \{a_3, a_5, a_7, a_8\}, \mathbb{H} \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to \frac{211701818}{1118425347}, a_2 \to \frac{2}{23}, a_3 \to \frac{4}{27}, a_4 \to \frac{3}{41}, a_5 \to \frac{1}{31}, a_6 \to \frac{5}{39}, a_7 \to \frac{16}{69}, a_8 \to \frac{12}{109}\}\}$

```
EvaluateProbability[{Pr[ℍ | 𝔼1], Pr[ℍ | 𝔼2], s[ℍ, 𝔼1], s[ℍ, 𝔼2]}, MODEL7] // N
```

```
{0.666543, 0.654647, 0.191741, 0.233022}
```

Here is a bizarre measure 𝛄, which satisfies ($\mathcal{R}$), but violates (8). This was discovered by Crupi *et al*. See their paper: "Towards a Grammar of Bayesian Confirmation", which can be downloaded from my Mathcamp webpage http://fitelson.org/mathcamp/

```
𝛄[ℍ_, 𝔼_] := Pr[ℍ | 𝔼]^2 - Pr[ℍ]^2;
```

This model shows that γ *violates* (8):

```
MODEL8 = PrSAT[{

    Pr[𝔼 | ℍ1] > Pr[𝔼 | ℍ2],
    Pr[𝔼 | ¬ ℍ1] == Pr[𝔼 | ¬ ℍ2],
    𝛄[ℍ1, 𝔼] < 𝛄[ℍ2, 𝔼]

   }, SearchAttempts → 10, Probabilities → Regular]
```

$$\left\{\{\mathbb{E}\to\{a_2,a_5,a_6,a_8\},\mathbb{H}1\to\{a_3,a_5,a_7,a_8\},\mathbb{H}2\to\{a_4,a_6,a_7,a_8\},\Omega\to\{a_1,a_2,a_3,a_4,a_5,a_6,a_7,a_8\}\},\left\{a_1\to\frac{50\,999\,038\,403}{426\,712\,555\,305},a_2\to\frac{14}{37},a_3\to\frac{118\,434\,037}{126\,860\,489\,415},a_4\to\frac{1}{27},a_5\to\frac{5}{33},a_6\to\frac{9}{29},a_7\to\frac{1}{999},a_8\to\frac{1}{785}\right\}\right\}$$

```
PrSAT[{

   Pr[𝔼 | ℍ1] > Pr[𝔼 | ℍ2],
   Pr[𝔼 | ¬ ℍ1] < Pr[𝔼 | ¬ ℍ2],
   J[ℍ1, 𝔼] < J[ℍ2, 𝔼],
   Pr[ℍ1] == 2 Pr[ℍ2]

  }, SearchAttempts → 10, Probabilities → Regular]
```

$$\left\{\{\mathbb{E}\to\{a_2,a_5,a_6,a_8\},\mathbb{H}1\to\{a_3,a_5,a_7,a_8\},\mathbb{H}2\to\{a_4,a_6,a_7,a_8\},\Omega\to\{a_1,a_2,a_3,a_4,a_5,a_6,a_7,a_8\}\},\left\{a_1\to\frac{28\,502}{177\,859},a_2\to\frac{4}{37},a_3\to\frac{280\,153}{865\,260},a_4\to\frac{7}{46},a_5\to\frac{13}{90},a_6\to\frac{1}{19},a_7\to\frac{1}{22},a_8\to\frac{1}{76}\right\}\right\}$$

We can verify that γ *satisfies* ($\mathcal{R}$), as follows:

```
PrSAT[{

   Pr[ℍ | 𝔼] > Pr[ℍ] && 𝛄[ℍ, 𝔼] ≤ 0 ||
    Pr[ℍ | 𝔼] == Pr[ℍ] && 𝛄[ℍ, 𝔼] ≠ 0 || Pr[ℍ | 𝔼] < Pr[ℍ] && 𝛄[ℍ, 𝔼] ≥ 0

  }, BypassSearch → True]

{}
```

My Hint about exercise (ii)

```
FullSimplify[(x - y)/(x + y) == Tanh[1/2 Log[x/y]]]

True
```

Both **`tanh`** and **`log`** are increasing functions. And, *increasing functions preserve order*.

Best Probability Bounds

`PrSAT` has a built-in function **`PrRange`** for computing best possible bounds on a probability expression, subject to (arbitrary) *numerical* constriants. Here are some examples:

Modus ponens:

In[63]:= $\text{PrRange}\left[\text{Pr}[B], \text{Pr}[B \mid A] == \frac{1}{2} \,\&\&\, \text{Pr}[A] == \frac{1}{3}\right]$

Out[63]= $\left\{\frac{1}{6}, \frac{5}{6}\right\}$

Constraints don't have to be equalities:

In[72]:= $\text{PrRange}\left[\text{Pr}[B], \text{Pr}[B \mid A] < \frac{1}{4} \,\&\&\, \text{Pr}[A] > \frac{1}{3}\right]$

Out[72]= $\left\{0, \frac{3}{4}\right\}$

Modus tollens:

In[75]:= $\text{PrRange}\left[\text{Pr}[\neg A], \text{Pr}[B \mid A] == \frac{1}{4} \,\&\&\, \text{Pr}[\neg B] == \frac{1}{3}\right]$

Out[75]= $\left\{\frac{5}{9}, 1\right\}$

Denying the antecedent:

In[76]:= $\text{PrRange}\left[\text{Pr}[\neg B], \text{Pr}[B \mid A] == \frac{1}{4} \,\&\&\, \text{Pr}[\neg A] == \frac{1}{3}\right]$

Out[76]= $\left\{\frac{1}{2}, \frac{5}{6}\right\}$

We can also solve analytically for general symbolic best possible bounds. Here are the bounds for *modus ponens*:

In[77]:= `MPCons = AlgebraicForm[Pr[B | A] == x && Pr[A] == y, {A, B}]`

Out[77]= $\frac{a_4}{a_2 + a_4} == x \,\&\&\, a_2 + a_4 == y$

In[78]:= `PrB = AlgebraicForm[Pr[B], {A, B}]`

Out[78]= $a_3 + a_4$

In[91]:= $\text{BackCons} = 0 < a_2 < 1 \,\&\&\, 0 < a_3 < 1 \,\&\&\, 0 < a_4 < 1 \,\&\&\, a_2 + a_3 + a_4 \le 1;$

In[95]:= $\text{MPMax} = \text{Maximize}[\{\text{PrB}, \text{MPCons} \,\&\&\, \text{BackCons}\}, \{a_2, a_3, a_4\}, \text{Reals}][[1]]$

Out[95]= $\begin{cases} 1 - y + x\,y & 0 < y < 1 \,\&\&\, 0 < x < 1 \\ -\infty & \text{True} \end{cases}$

In[96]:= `FullSimplify[MPMax, 0 < y < 1 && 0 < x < 1]`

Out[96]= $1 + (-1 + x)\,y$

In[97]:= $\text{MPMin} = \text{Minimize}[\{\text{PrB}, \text{MPCons} \,\&\&\, \text{BackCons}\}, \{a_2, a_3, a_4\}, \text{Reals}][[1]]$

Out[97]= $\begin{cases} x\,y & 0 < y < 1 \,\&\&\, 0 < x < 1 \\ \infty & \text{True} \end{cases}$

In[98]:= `FullSimplify[MPMin, 0 < y < 1 && 0 < x < 1]`

Out[98]= $x\,y$

And, here are the bounds for *modus tollens*:

In[99]:= `MTCons = AlgebraicForm[Pr[B | A] == x && Pr[¬ B] == y, {A, B}]`

Out[99]= $\frac{a_4}{a_2 + a_4} == x \,\&\&\, 1 - a_3 - a_4 == y$

```
In[101]:= PrNotA = AlgebraicForm[Pr[¬ A], {A, B}]

Out[101]= 1 - a2 - a4

In[102]:= MTMax = Maximize[{PrNotA, MTCons && BackCons}, {a2, a3, a4}, Reals][[1]]

Out[102]= { 1    0 < y < 1 && 0 < x < 1
          { -∞   True

In[103]:= FullSimplify[MTMax, 0 < y < 1 && 0 < x < 1]

Out[103]= 1

In[104]:= MTMin = Minimize[{PrNotA, MTCons && BackCons}, {a2, a3, a4}, Reals][[1]]

Out[104]= { (-1+x+y)/(-1+x)   (0 < y < 1 && x == 1 - y) || (0 < y < 1 && 0 < x < 1 - y)
          { (-1+x+y)/x        0 < y < 1 && 1 - y < x < 1
          { ∞                 True

In[105]:= FullSimplify[MTMin, 0 < y < 1 && 0 < x < 1]

Out[105]= { 1 + y/(-1+x)      x + y ≤ 1
          { (-1+x+y)/x        True

In[106]:= FullSimplify[MTMin, 0 < y < 1 && 0 < x < 1 && x + y ≤ 1]

Out[106]= 1 + y/(-1 + x)

In[107]:= FullSimplify[MTMin, 0 < y < 1 && 0 < x < 1 && x + y > 1]

Out[107]= (-1 + x + y)/x
```